
Les buts principaux qui ont présidé à la conception du langage PASCAL II et du Compilateur CPAS sont :

· Disposé d’un Pascal à la fois vraiment portable sur toute machine moderne et raisonnablement efficace sur le plan des performances de calcul du code généré.

· Disposé au travers de ce compilateur d’un langage puissant et rigoureux tant pour le calcul scientifique que pour la création l’informatique en générale aussi indépendant que possible d’une architecture donnée.

En conséquence il se doit de permettre de traiter de façon aussi souple que possible les entrées-sorties, les événements les objets mathématiques structurés tel que les complexes, les matrices … .

Par PASCAL étendu, nous entendons que ce langage inclue les caractéristiques du PASCAL original (dit « PASCAL ZURICH » – Norme ANSI/IEEE770X3.97-1983 (ANSI) et ISO 7185-1983(E) (ISO)) ainsi que des extensions empruntées pour une large part au PASCAL Digital (suivant partiellement la norme X3J9/IEEE P770 et IOS IEC/JTC1/SC22/WG2) qui été utilisé sur les systèmes d’exploitation OPEN-VMS des VAX et ALPHA ainsi qu’aux langages ADA et PL/1. La puissance de ce langage lui permet alors d’être utilisé comme un langage « presque objet » pour lequel les procédures/fonctions ainsi que les opérateurs peuvent être surchargés pour manipuler les types décrits par le développer. Ce langage inclue aussi un contrôle sur les événements pouvant intervenir durant l’exécution comme une erreur de calcul ou une interaction avec d’autre processus.

Le compilateur CPAS constitue une « réalisation » de ce langage pouvant fonctionner sur tout système UNIX/POSIX ainsi que sur les systèmes Windows (Marque déposée Microsoft) par l’intermédiaire de l’environnement CYGWIN, logiciel libre disponible sur les sites www.redhat.com. Ce compilateur fourni un code C très proche d’un langage d’assemblage et néanmoins performant qui est ensuite lu par un compilateur C pour obtenir le langage machine. Les compilateur gcc (Windows/Cygwin et Linux), et CC des stations HP (HPUX), Silicon-Gaphique (IRIX) et Digital (True Unix/OSF1, Linux ou OPEN-VMS Alpha et Vax). De plus d’autres compilateurs peuvent facilement être utilisés sur pratiquement tous les systèmes de type UNIX/POSIX.

Enfin si CPAS génère du C en sortie il n’est en aucune manière un traducteur de langage mais un vrai compilateur qui pourrait tout aussi bien générer un code objet ou un code d’assembleur en ne changeant que le module de génération de code ou encore en ajoutant un autre.

Le compilateur CPAS une réalisation de PASCAL II

I - Les limitations de CPAS par rapport au PASCAL Zurich

Les limitations par rapport au PASCAL Zurich (Pascal original) sont peu nombreuses et sont essentiellement liées à la conception du compilateur CPAS.

I.1 - Les procédures standard dont les noms sont des mots clés (réservé et ne peuvent être utilisés pour d’autre propos) :

read, readln, write, writeln,

open, reset, rewrite, append,

new.

On ne peut donc pas redéfinir à ce niveau les entrés sorties standards

I.2 - Les nouveaux mots clés qui permettent de gérer des extensions du langage (plusieurs d’entre eux sont reconnus mais non gérés par CPAS) :

readv, writev,

in_var, out_var,

private, access,

user_statement,

module, use,

external, standard, inline,

loop, exit, otherwise, return,

%define, %include, %chaine, %pragma, %eof, %if, %ifdef.

I.3 – Les restrictions :

A - Sur les branchements sortant de procédures et/ou fonctions internes. Un tel goto ne peux en aucun cas faire un branchement qui aurait pour effet d’exécuter des instructions internes à une instruction structurée telle que begin … end, repeat … until, même si l’appel du sous programme est la conséquence d’une instruction dans cette même instruction structurée. En fait cette restriction qui est une conséquence de la réalisation de CPAS dont le résultat est un code C et non pas du code assembleur.

Exemple :

Procedure P1;

…

label 100, 200;

…

  procedure P2

  var

    machin, truc: Boolean;

  begin

    …

    if machin then goto 100;  (* Branchement possible*)

    …

    if truc then goto 200;    (* Branchement illegal *)

  end { P2 };

begin { P1 }

100:

  repeat

200:

  until k = 0;

end {P2 };

B - La procédure standard DISPOSE ne supporte qu’un seul paramètre, le pointeur de la zone mémoire à libérer. Les paramètres supplémentaires ne servent à rien et doivent être supprimés ou mis en commentaire.

II - Les macros instructions et les packages

Les instructions %pragma, %include, %chaine et %eof permettent de partagé le code d’un même programme sur plusieurs fichiers sources et de modifié les options de compilation de certaines parties du code. %include <file_spécification> …; agit comme si l’on avait inclue les lignes du fichier spécifié à la place de l’instruction tandis que %chaine <file_spécification> …; provoque l’abandon de la lecture sur fichier source courant pour continuer cette lecture sur le fichier indiqué et que %eof se comporte comme une fin de fichier source avant sa fin réelle. %pragma …; permet de modifier les options courantes de fonctionnement du compilateur comme (entre autre) valider ou invalider le contrôle des bornes lors de l’affectation de variable intervalle ou lors de l’accès à un élément de tableau.

De futures versions de CPAS introduiront la possibilité de définir des macros et des compilations conditionnelles à l’aide des mots clés %define, %if et %ifdef un peu comme il est possible de le faire en C/C++ et en PL/1.

Le mot clé module permet de définir un module sans programme principal et permet de réalisé des programmes assemblés à partir de plusieurs module qui peuvent eux même être organisé en librairie. Les attributs de modules et/ou programme inherit et environment (syntaxe venant du Dec-Pascal), ainsi que la directive use sont réservés pour de futures versions de CPAS afin de faciliter l’écriture de programme avec des modules organisés en package comme cela existe en ADA. En attendant cela se fait avec des fichiers d’environnement que l’on inclut, un peu à la manière du C. Des facilités existent tel que les qualificatifs global et external :

[global] var …et [external] var … qui définissent comme globales et externes respectivement toute les variables définies après le mot clé var. Ceci s’applique aussi aux procédures et fonctions. Enfin un même objet préalablement défini externe peut être redéfini global ensuite ce qui permet de définir global des procédures d’un module qui utilise un fichier d’environnement où toutes les constantes, types, variable, fonctions et procédures que l’on veut rendre visibles de l’ensemble du programme sont définis. C’est par ce moyen que sont actuellement définis les environnements liés aux bibliothèques standards et complémentaires. Enfin les qualificatifs global et external peuvent spécifier un nom global (une chaîne de caractère) dans la mesure où il concerne un seul objet (variable, fonction ou procédure). Ainsi

var

  theta, 

  [global ‘Lambda’] wave_length: real;

indique que la variable wave_length est définie globalement avec un nom externe qui est Lambda (avec lettres minuscules et majuscules distinguées) ce qui permet de même de définir des objets accessibles aussi par un programme en C ou C++. Ce nom spécifié ainsi est celui qui sera reconnu comme symbole global par l’éditeur de lien. De la même manière, cela peut être fait aussi pour les procédures et fonctions sachant que dans le cas externe le qualificatif external n’est pas utile car le mot clé external peut être suivi de la chaîne de caractère, ainsi :

[global ‘Ma_Fonction’] fonction F1( v: real ): integer;

begin

  …

  F1 := …

end;

défini F1 comme de porté global sous le nom externe « Ma_Fonction » tandis que :

function sa_fonction( v: real ): integer; external ‘MA_Fonction’;

défini cette même fonction F1 sous le nom Pascal « sa_fonction » comme externe et portant le nom global « Ma_Fonction ».

III - Les attributs et qualificatifs

Nous venons de voir les qualificatifs (toujours entre des crochets […] ) external et global. Le qualificatif static permet de définir des variables statiques qui sont alloués par le compilateur et non pas à l’exécution. Il est à noter que les variables du programme principal ainsi que celle déclaré dans un corps de module (mais pas dans un sous-programme) sont en fait toujours statiques pour CPAS.

Les attributs byte, word et long servent à préciser devant un type peuvent être utilisé pour indiquer s’il doit occuper un octet [byte], un mot de 16 bits [word] ou un mot de 32 bits [long] comme cela se fait dans le Pascal Dec. Ainsi

code: [byte] 1..10;

précise qu’il ne faut utiliser qu’un seul octet pour la variable code qui sans cela se verrait attribuer une taille de 4 octets (soit 32 bits). Ceci peut être aussi spécifié par l’usage, après le type, de la syntaxe « use <size_en_bits> [<alignement_en_bits>] ».

D’autres qualificatifs peuvent être utilisés comme, par exemple, [readonly] ou [optional] pour un paramètre formel de procédure qui ne doit pas être modifié ou encore un paramètre par référence (définie avec le mot clé var) qui peut ne pas être fourni lors de l’appel et dont l’adresse est alors nil ce qui peut être testé en utilisant l’attribut address ainsi :

If parm"address <> nil then

Ceci nous montre aussi notre premier usage d’attribut d’un objet (syntaxe inspirée de ADA). Les attributs les plus utilisés sont :

size
qui renvoie la taille en octet de l’objet (type ou variable),

address
qui renvoie l’adresse de la variable (type $wild_pointer),

first
qui renvoie la plus petite valeur permise d’un type dénombrable,

last
sa valeur la plus grande,

cardinality
la cardinalité d’une variable/d’un type ensemble

epsilon
le plus petit nombre > 0.0 d’un type real (single or double),

large
le plus grand nombre d’un type réel (single ou double).

IV - Les types paramétrés

Les types paramétrés sont des types plus souples que les types ordinaires de Pascal car ils permettent de définir des objets de taille variable. L’exemple le plus simple est celui du type standard string qui est défini (lors de l’initialisation du compilateur CPAS) par la définition suivante ;

type

  short_natural  = 1..255   use 8, 8;    { Natural  on a byte }

  short_unsigned = 0..255   use 8, 8;    { Unsigned on a byte }

  byte = short_unsigned;                 { Byte definition }

  string( capacity: short_natural := 255 ) = record

    length: byte;

      body: array[1..capacity] of char

  end;

Une procédure ou une fonction qui spécifiera un paramètre formel s du simple type « string » acceptera n’importe quelle string et aura accès à a sa valeur allouée en caractère par la référence « s.capacity ». capacity est le paramètre formel unique du type string et sa valeur par défaut est 255.

Trois restrictions importantes doivent être faites :

1. Une variable de type var ne peut avoir qu’une taille prédéterminée par le compilateur (en spécifiant par exemple « string( 32 ) »). Une taille vraiment variable sera obtenue au moyen d’une allocation par new ainsi :

  var

    P: ^string;

  …

  begin

    READ( n );

    NEW( p, n );

    …

  end

2. Dans un record, seul le dernier champ peut avoir une taille paramétrée. Tous les autres objets doivent avoir une taille fixe.

3. Les record avec case sont actuellement incompatibles avec les types paramétrés mais seront possible dans des version future de CPAS à la condition que le sélecteur du case soit un des paramètres du type (comme en ADA).

4. Aucun fichier ne peut avoir comme type d’élément un type paramétré actuellement (tel que « file of string » ou « file of matrix(5, 5) » car les descripteurs nécessaires ne seraient pas gérés par la bibliothèque standard.

Il est possible de définir des objets du type relativement complexes comme :

  type

    vecteur( nv: integer) = array[1..nv] of real;

    matrix(n, m: integer) = array[1..n] of vecteur( m );

    menu( nc, cap: byte) = array[1..nc] of string( cap );

  …

  var

    m1: matrix( 4, 3 );

    p: ^matrix;

    nx, ny : integer ;

  ...

  begin

    READ( nx, ny );

    NEW( p, nx, ny );

    for i := 1 to p^.n do

      for j := 1 to p^[i].nv do

        p^[i,j] := i*j;

  ...

  end

V - Les constantes structurées

Il est possible de fournir des valeurs initiales lors de la déclaration de variables ainsi « <identificateur> : <type> := <valeur_initiale_constante> ; ».

Pour un type structuré, record et/ou array la constante structurée consiste en une suite de valeurs séparées par des virgules et encadrées par des crochets ainsi :

  Var

    V1: vector( 3 ) := [0.51, -0.25, 0.0];

    p: ^matrix;

  ...

  begin

  ...

    V1 := 

    P^ := matrix[ [  1.0,  0.0,  0.0, 0.5],

                  [  0.0, -1.0,  0.0, 0.5],

                  [  0.0,  0.0, -1.0, 0.0] ];

  ...

Nota : dans le cas des chaînes de caractères, il est équivalent d’écrire :

s: string( 10 ) := ‘et voila’;

ou

s: string( 10 ) := [ 8,‘et voila’];

ou

s:string(10):=[8,[‘e’,‘t’,‘ ’,‘v’,‘o’,‘i’,‘l’,‘a’, ‘ ’,‘ ’] ];

à ceci prêt que la première forme mettra CHR( 0 ) dans les caractères non fournis alors que les 2 autres formes compléteront à dix caractères avec des blancs comme pour une affectation entre deux tableaux de caractères.

Il faut remarquer que les paramètres ne doivent pas être précisés dans la constante mais dans la déclaration de type ou en paramètre de la procédure NEW. Pour les record(s) avec case de Pascal il faudra par contre toujours préciser la valeur du sélecteur de case qu’il soit un vrai champ de la structure ou un simple type.

VI - Les surcharges des opérateurs et des sous-programmes

Nous entendons par surcharge la possibilité de redéfinir des procédures, des fonctions ou des opérateurs pour des types d’objets définis par le dévellopeur. Par exemple on pourra définir comment écrire une matrice en redéfinissant la procédure WRITE$OBJECT qui est utilisés par les instructions write, writeln et writev pour écrire les objets de types standards :

  procedure WRITE$OBJECT( in_var mat: matrix; f, d, n: interger := 0 );

..{ Usage de mat.m et mat.n pour accéder aux dimensions de la matrice.}

  ...{

ce qui pourra être utilisé de cette façon dans une instruction writeln :

  WRITELN( ‘ Ma matrice est ‘, mat:12:5 );

  ...

Il est bien sûr possible de faire la même chose avec la procédure générique de lecture READ$OBJECT qui est invoquée dans les instructions read, readln et readv.

De façon très générale on peut définir dans le même bloc autant de procédure ou fonction avec le même nom dans la mesure où la liste de leurs paramètres formels sont différentes (tous les objets globaux sont considérés comme appartenant à l’un des blocs standard ou global). La surcharge des opérateurs se fera avec les mêmes règles ainsi :

type

  complex = record

             rp, ip: double

            end;

  function * (z1, z2: complex) : complex;

  var

    r : complex;

  begin

    r.rp := z1.rp*z2*rp – z1.ip*z2.ip;

    r.ip := z1.rp*z2.ip + z1.ip*z2.rp;

    return r

  end;

...

défini le type complexe ainsi que la multiplication de deux complexes. On notera l’usage de l’instruction return qui permet de renvoyer sa valeur dans une fonction dont le nom est un opérateur. Il est ainsi possible de créer des packages pour les calculs sur des nombres complexes, des matrices éventuellement elles-mêmes complexes, etc … .

Tous les opérateurs Pascal peuvent être surchargés, même l’opérateur d’assignation := . Les contraintes à respecter sont les suivantes :

Procedure := ( <var/out_var> dst <type>; <by_value/in_var> <type> ) ;

Function <bope> (<2_formels_par_valeur (ou in_var)>): <resulting type>;

Function <uope> (<1_formel_par_valeur (ou in_var)>): <resulting type>;

avec <bope> un opérateur binaire :

 +, -, *, /, **,  (* Opérateurs numériques avec ** pour puissance

                   et + pour Union, * pour intersection

                   – pour suppression avec les ensembles (set of) *)

 and, or, xor,    (* Opérateurs logiques sur les booleans *)

 in, =, <>, <,    (* teste d’appartenance (set), égales, non égales *),

 <, <= >=, >,     (* autres comparaisons et test d’inclusion (set) *)

 ||               (* l’opérateur de concaténation de chaîne *)

avec <uope> un opérateur unaire :

 +, -, not        (* + unaire, - negation ou complement (set),

                     non logique *)

L’usage les mots clés return et function permet aussi quelques facilités, comme accédé à des champs particuliers de la valeur à retourner, comme, par exemple dans une fonction retournant une chaîne de caractère « function.capacity » pour obtenir la capacité de la chaîne retournée (valeur à lecture seule) ou pour forcer un champ particulier de la valeur à retourner comme « return.length := 0 » pour renvoyer une chaîne vide. Ainsi la procédure multiplication de deux complexes aurait pu être écrite :

function * (z1, z2: complex) : complex;

begin

  return.rp := z1.rp*z2*rp – z1.ip*z2.ip;

  return.ip := z1.rp*z2.ip + z1.ip*z2.rp;

end;

VII - Les extensions sur les structures de contrôle

VII.1 Les étiquettes (ou labels) de PASCAL II et CPAS.

La restriction imposée par Pascal Zurich à l’usage des étiquettes (label) qui doivent être des nombres entiers est de nature, elle-même, à rendre le programme moins lisible. Aussi Pascal II (et CPAS) supporte t-il des étiquettes de la forme <identificateur>: qui sont beaucoup plus claires. Cela ne signifie pas que nous encourageons l’usage des instructions goto qui, par nature, nuisent grandement à la lisibilité des codes sources. Aussi prônons-nous un usage modéré des instructions goto.

Il faut distinguer deux sortes de goto. Ceux qui permettent de faire des branchements à l’intérieur d’un unique bloc (procédure ou fonction), et ceux qui permettent de sortir d’un tel bloc. Dans le premier cas, le compilateur n’exige pas la déclaration de l’étiquette mais vérifie la validité du branchement qui ne doit en aucun cas provoquer l’entrée illégale dans une instruction structurée tel que begin … end, repeat … until ou encore loop … end. Dans le second cas l’étiquette doit être déclarée par l’instruction label dans le bloc où elle se trouve, ce qui la rend accessible aux instructions goto des blocs plus internes.

VII.2 Les boucles de PASCAL II et CPAS.

Le mot clé loop permet de définir des boucles sans conditions d’arrêt qui peuvent être contrôlée par l’instruction exit ce qui permet de réaliser des boucles avec plusieurs point de sortie. Ainsi :

loop

  ...

exit if <cond_1>; (* sortie n° 1 si cond_1 = true *)

  ...

exit if <cond_2>; (* sortie n° 2 si cond_2 = true *)

  ...

end;

Naturellement l’instruction exit peut aussi être utilisée dans n’importe quelle autre forme de boucle : for, while, repeat … until.

Il est possible de plus (comme en ADA) d’étiqueter une boucle ce qui permet de spécifier dans l’instruction exit la boucle que l’on termine. Ainsi dans l’exemple suivant la deuxième instruction exit stop les deux boucles à la fois :

main_loop:

  while <cond_1> do

  begin

    ...

    loop

      ...

    exit if <cond_1>;         (* Sortie de la boucle loop si cond_1 *)

    ...

  exit main_loop if <cond_2>; (* Sortie de la boucle while si cond_2 *)

    ...

    end ;

    ...

  exit if <cond_3> ;          (* Sortie de la boucle while si cond_3 *)

    ...

end;

Ces facilitées permettent pratiquement de ne jamais faire de goto. Ceux-ci seront alors réservés à des situations exceptionnelles. Malgré tout, le goto qui permet la sortie d’un contexte de bloc à la suite d’une erreur est par contre extrêmement utile pour réaliser des programmes robustes et ne peut pas être évité dans ce cas. L’exemple qui suit nous permet en passant de voir que l’on peut faire suivre les mots clés end terminant une procédure une fonction, un programme ou un module par l’identificateur de ce même bloc. Cela permet au compilateur de vérifier l’imbrication des structures du programme. On y voit aussi l’usage de la procédure standard ESTABLISH qui permet de rendre actif une fonction «gestionnaire d’événement », le dit événement étant enclenché automatiquement soit par une erreur standard, soit par l’appel de la fonction ERROR avec un numéro >= 10000 ou < 0 (pour faire la distinction avec une erreur standard). Cette procédure est accessible par la déclaration suivante :

procedure ERROR( nerr: integer ); external ‘PAS__ERROR’;

Cette déclaration est actuellement nécessaire car CPAS ne considère pas son appel comme standard, le principal usage étant le traitement des erreurs.

Exemple es0.pas du kit CPAS qui teste le TRAP de l’erreur ‘Floatting overflow’  :

program ESSAI_0;

label et_stop;

type

  real = single;

var

  i,j: integer;

function err_hdl( ierr: cc__int ): cc__int;

begin { Fontion Handler d’erreur de débordement flottant }

  WRITELN( err, 'L''erreur # ', ierr, ' a ete detectee.' );

  goto et_stop;

  err_hdl := 1

end err_hdl;

function FAC( r$a: real ): real;

begin

  if ABS(r$a - 1.0) < 0.1 then FAC := 1.0

                          else FAC := r$a*FAC( r$a – 1 )

end FAC ;

begin { ESSAI_0 }

  ESTABLISH( err_hdl ); }

  i := 1;

  loop

    WRITELN( ' i = ', i:6, ', fac(i) =', FAC( REAL( i ) ):10:5 );

    i := i + 1;

  exit if i > 500;

  end { loop };

et_stop:

end ESSAI_0.

VIII - La librairie standard

Voici une liste des procédures et fonctions disponibles :

GET_LOGICAL
Obtient la valeur d’un symbole logique ou d’une variable d’environnement.

SET_LOGICAL
Définie ou détruit une nouvelle variable d’environnement.

RESET
Ouverture d’en fichier en, lecture,

REWRITE
Ouverture d’un fichier en écriture,

APPEND
Ouverture d’un fichier et mode « d’allongement »,

OPEN
La procédure générale d’ouverture de fichier (sur disque ou.des pipes).

SEEK
Positionnement sur un fichier ouvert en mode Direct (par OPEN).

CLOSE
Fermeture d’un fichier.

GET_DEF_DIR
Renvoie le directory par défaut du moment.

CHANGE_DIRECTORY
Change de directory par défaut.

SYS_SPAWN
Envoie une commande à l’interpréteur de commande courant (shell).

RUN PROCESS
Ecrase le programme courant par un programme spécifié en paramètre (c’est la fonction EXEC de Unix).

CREATE_PROCESS
Crée un processus enfant du processus courant.

WAIT_PROCESS
Attend la fin d’exécution d’un processus donné.

CREATE_PIPE
Crée une pipe qui peut-être ensuite géré par des fichiers avec l'instruction OPEN.

EOF et EOLN
Fin de fichier et fin de ligne du Pascal Traditionnel.

CURRENT_EOF

CURRENT_EOLN
Fin de fichier et fin de ligne du fichier sélectionné par un READ/READLN/WRITE/WRITELN.

READ$OBJECT et WRITE$OBJECT procédure de lecture et d’écriture de tous les types de base (pour les surchargés).

PAS$$SELECT_INPUT et PAS$$SELECT_OUTPUT


Procédures de sélection de fichier ouvert (pour les surchargés).

PAS$$READ_EOLN et PAS$$WRITE_EOLN


Procédure d’écriture ou de lecture de fin de ligne (pour les surchargés)

GET et PUT
Procédure de base d’entrée et sortie du PASCAL.

TTY_FILE
Renvoie true si le fichier spécifié et ouvert sur un terminal (ou assimilé).

TTY_CLR_EOF
Permet de désarmer le drapeau fin de fichier sur un fichier terminal.

FILE_STATUS
Renvoie le mot d’état du fichier spécifié.

FILE_SPECIFICATION
Renvoie la chaîne de caractère de spécification du fichier ouvert donné en paramètre.

PAGE
Pour générer un saut de page sur un fichier texte en écriture.

OPEN_DIR, CLOSE_DIR et READ_DIR


Pour pouvoir lire des directories.

MAKE_DIR et REMOVE_DIR


Pour créer et détruire des directories.

SUBSTR
Pour extraire une sous-chaîne d’une chaîne de caractères.

INDEX
Pour localiser une sous-chaîne dans une chaîne de caractères.

SET_CASE
pour mettre en Majuscule/minuscule une chaine de caractères.

STR_MATCH
Pour comparer deux chaînes.

STRING_LOCATE_SEP
Pour localiser des séparateurs dans une chaîne.

RANDOM
Pour générer des nombres pseudo-aléatoires.

PASCAL_EXIT
Pour terminer l’exécution du programme avec un code d’erreur on de sucés (0 en général).

CLOCK
Pour mesurer le temps d’exécution CPU.

DATE et TIME
Pour obtenir la date et l’heure.

FILE_RENAME
Pour renommer un fichier.

FILE_REMOVE
pour effacer un fichier.

PATH_SEARCH
Pour rechercher dans un PATH (variable d’environnement avec une liste de directory à examiner).

FILE_ACCESS_CHECK
Pour tester l’existence et/ou les droits d’accès à un fichier

SET_TIMER et WAIT_TIMER


Pour gérer l’horloge et se synchroniser avec.

SLEEP
Pour attendre un délai donné.

ESTABLISH et REVERT


Pour installer et désinstaller une fonction gestionnaire d’événement.

GET_RTL_MESSAGE
Pour retrouver la chaîne donnant la signification d’un message d’erreur d’exécution standard.

IX - Les librairies complémentaires

Les librairies basic_lst, basic_err et basic_src (inclusent dans libcpas) :

Il s’agit de librairie permettant de gérer la lecture d’un code source (fichier de commande à interpréter, programme dans une langage donné…) d’en fournir les caractères au programme utilisateur, d’en éditer les éventuels messages d’erreur ainsi que des fichiers listing bien formatés avec gestion des fins de pages, des paragraphes et numérotation des pages avec un en-tête, un titre et un sous-titre. Différents formats de commentaires sont également traités. Ces modules sont utilisés par le compilateur cpas et l’interpréter de commande cpsh.

Cpas_d__sdir1 et cpas_d__sdir2, qui sont des modules de gestion des dossiers (directories ou catalogues) qui permet de parcourir les arborescences de fichiers.

La librairie DRAW pour réaliser des sorties graphiques via OPENGL :

Cette librairie utilisable en CPAS, en C et C++ ainsi qu’en fortran et basée sur une version de la bibliothèque V (en C++, de Bruce Wampler, URL Web : www.objectcentral.com) avec un serveur graphique (en C++), et une librairie en C/F77/CPAS d’interface.

X - Les outils complémentaires

Le générateur de fichier de message (fichiers séquentiels indexés) gen_msgfile qui à partir d’un fichier ASCII crée le fichier de message.

L’interpréteur cpsh qui est utilisé pour les installations de produit CPAS par la commande « cpsh install ».

XI – Quelques applications disponibles en kit cpsh

- Le logiciel de cristallographie MXD et ces programmes associés.

- L’interpréteur de LISP étendu ELISP avec l’application DPU pour les minimisation à partir des formules algébriques (méthodes de moindres carrés) avec interface graphique DRAW/OpenGL.



